<commit_before># info.nix -- install texinfo, set INFOPATH, create `dir` file

# This is a helper for the GNU info documentation system. By default,
# the `info` command (and the Info subsystem within Emacs) gives easy
# access to the info files stored system-wide, but not info files in
# your ~/.nix-profile.

# We set $INFOPATH to include `/run/current-system/sw/share/info` and
# `~/.nix-profile/share/info` but it's not enough. Although info can
# then find files when you explicitly ask for them, it doesn't show
# them to you in the table of contents on startup. To do that requires
# a `dir` file. NixOS keeps the system-wide `dir` file up to date, but
# ignores home-installed packages.

# So this module contains an activation script that generates the
# `dir` for your home profile. Then when you start info (and both
# `dir` files are in your $INFOPATH), it will *merge* the contents of
# the two files, showing you a unified table of contents for all
# packages. This is really nice.

{ config, lib, pkgs, ... }:

with lib;
with import ../lib/dag.nix;

let
  cfg = config.programs.info;

  # Indexes info files found in this location
  homeInfoPath = "$HOME/.nix-profile/share/info";

  # Installs this package -- the interactive just means that it
  # includes the curses `info` program. We also use `install-info`
  # from this package in the activation script.
  infoPkg = pkgs.texinfoInteractive;

in

{
  options = {
    programs.info = {
      enable = mkEnableOption "GNU Info";

      homeInfoDirLocation = mkOption {
        default = "$HOME/.cache/info";
        description = ''
          Directory in which to store the info <filename>dir</filename>
          file within your home.
        '';
      };
    };
  };

  config = mkIf cfg.enable {
    assertions = [{
      assertion = config.home.sessionVariableSetter != "pam";
      message = ''
        The info module does not work with PAM as a session variable setter.
      '';
    }];

    home.sessionVariables.INFOPATH =
      "${cfg.homeInfoDirLocation}\${INFOPATH:+:}\${INFOPATH}";

    home.activation.createHomeInfoDir = dagEntryAfter ["installPackages"] ''
      $DRY_RUN_CMD mkdir -p "${cfg.homeInfoDirLocation}"
      $DRY_RUN_CMD rm -f "${cfg.homeInfoDirLocation}/dir"
      if [[ -d "${homeInfoPath}" ]]; then
        find -L "${homeInfoPath}" \( -name '*.info' -o -name '*.info.gz' \) \
          -exec $DRY_RUN_CMD ${infoPkg}/bin/install-info '{}' \
          "${cfg.homeInfoDirLocation}/dir" \;
      fi
    '';

    home.packages = [infoPkg];
  };
}
<commit_msg>info: use XDG_CACHE_HOME if defined<commit_after># info.nix -- install texinfo, set INFOPATH, create `dir` file

# This is a helper for the GNU info documentation system. By default,
# the `info` command (and the Info subsystem within Emacs) gives easy
# access to the info files stored system-wide, but not info files in
# your ~/.nix-profile.

# We set $INFOPATH to include `/run/current-system/sw/share/info` and
# `~/.nix-profile/share/info` but it's not enough. Although info can
# then find files when you explicitly ask for them, it doesn't show
# them to you in the table of contents on startup. To do that requires
# a `dir` file. NixOS keeps the system-wide `dir` file up to date, but
# ignores home-installed packages.

# So this module contains an activation script that generates the
# `dir` for your home profile. Then when you start info (and both
# `dir` files are in your $INFOPATH), it will *merge* the contents of
# the two files, showing you a unified table of contents for all
# packages. This is really nice.

{ config, lib, pkgs, ... }:

with lib;
with import ../lib/dag.nix;

let
  cfg = config.programs.info;

  # Indexes info files found in this location
  homeInfoPath = "$HOME/.nix-profile/share/info";

  # Installs this package -- the interactive just means that it
  # includes the curses `info` program. We also use `install-info`
  # from this package in the activation script.
  infoPkg = pkgs.texinfoInteractive;

in

{
  options = {
    programs.info = {
      enable = mkEnableOption "GNU Info";

      homeInfoDirLocation = mkOption {
        default = "\${XDG_CACHE_HOME:-$HOME/.cache}/info";
        description = ''
          Directory in which to store the info <filename>dir</filename>
          file within your home.
        '';
      };
    };
  };

  config = mkIf cfg.enable {
    assertions = [{
      assertion = config.home.sessionVariableSetter != "pam";
      message = ''
        The info module does not work with PAM as a session variable setter.
      '';
    }];

    home.sessionVariables.INFOPATH =
      "${cfg.homeInfoDirLocation}\${INFOPATH:+:}\${INFOPATH}";

    home.activation.createHomeInfoDir = dagEntryAfter ["installPackages"] ''
      $DRY_RUN_CMD mkdir -p "${cfg.homeInfoDirLocation}"
      $DRY_RUN_CMD rm -f "${cfg.homeInfoDirLocation}/dir"
      if [[ -d "${homeInfoPath}" ]]; then
        find -L "${homeInfoPath}" \( -name '*.info' -o -name '*.info.gz' \) \
          -exec $DRY_RUN_CMD ${infoPkg}/bin/install-info '{}' \
          "${cfg.homeInfoDirLocation}/dir" \;
      fi
    '';

    home.packages = [infoPkg];
  };
}

{'fraction_non_alphanumeric': 0.11996124031007752, 'fraction_numerical': 0.0, 'mean_word_length': 2.9366895499618613, 'pattern_counts': {'":': 0, '<': 7, '<?xml version=': 0, '>': 7, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '3594201', 'n_tokens_mistral': 1612, 'n_tokens_neox': 1561, 'n_words': 560}